# $Id$
# Maintainer: Jan de Groot <jgc@archlinux.org>

pkgname=xorg-xmodmap
pkgver=1.0.9
pkgrel=1
pkgdesc="Utility for modifying keymaps and button mappings"
arch=('i686' 'x86_64')
url="https://xorg.freedesktop.org/"
license=('custom')
depends=('libx11')
makedepends=('xorg-util-macros')
groups=('xorg-apps' 'xorg')
source=(https://xorg.freedesktop.org/archive/individual/app/xmodmap-${pkgver}.tar.bz2{,.sig})
sha256sums=('b7b0e5cc5f10d0fb6d2d6ea4f00c77e8ac0e847cc5a73be94cd86139ac4ac478'
            'SKIP')
validpgpkeys=('4A193C06D35E7C670FA4EF0BA2FB9E081F2D130E') # Alan Coopersmith <alan.coopersmith@oracle.com>

build() {
  cd xmodmap-${pkgver}
  ./configure --prefix=/usr
  make
}

package() {
  cd xmodmap-${pkgver}
  make DESTDIR="${pkgdir}" install
  install -Dm644 COPYING "${pkgdir}/usr/share/licenses/${pkgname}/COPYING"
}
